

Bloomberg

MICROSOFT[®] EXCEL ADD-IN

AVAILABLE FOR EXCEL 2002 AND ABOVE

Function Reference

25 October 2007

Version: 1.0

Bloomberg

Bloomberg Functions & Calculation Overrides	1
BDP (Bloomberg Data Point)	1
Single Security with a Single Field	1
Single Security with Multiple Fields	2
Multiple Securities with Multiple Fields	3
BDS (Bloomberg Data Set)	4
BDH (Bloomberg Data History)	5
Historical Time Series	5
Intraday Historical Time Series	7
Overrides	8
Cell referencing	8
Returning a value to one cell	9
Cell referencing with multiple input cells/values	9
Absolute Referencing	10
Appendix A: Optional Arguments for BDH formulas	12

Bloomberg Formulas & Calculation Overrides

In the new Microsoft® Excel Add-in the Bloomberg functions have been consolidated and the optional parameters made more flexible. The new functions are:

BDP	BDP (Bloomberg Data Point) is for current data. This is the equivalent of BLP.
BDH	BDH (Bloomberg Data History) is for historical end of day and historical intraday data. This is the equivalent of BLPH, BLPSH and BLPI.
BDS	BDS (Bloomberg Data Set) is for large data sets/ bulk data. This is the equivalent of BLPB.

BDP (Bloomberg Data Point)

BDP (Bloomberg Data Point) returns the same data as the Market Data/Reference data option in the Data Wizard. This is static or real time current market data.

It is the equivalent of the BLP formula from the previous add-in. The previous method of automatic array creation when using multiple securities or fields is **no longer** permitted. Excel cell locking rules apply if using multiple fields/securities.

Syntax


=BDP (security, field)

Single Security with a Single Field

The single security with a single field cell reference, to be entered in cell B2, is:

=BDP (A2,B1)

where **A2** contains the security and **B1** contains the field mnemonic, as shown in Figure 1.



The screenshot shows an Excel spreadsheet with the following data:

	A	B	C
1		px last	
2	goog equity	=BDP(A2,B1)	
3			

The formula bar at the top shows the active cell (B2) containing the formula =BDP(A2,B1).

Figure 1: Single Security with a Single Field Example

The absolute reference of the above example, to be entered in cell B2, is:

=BDP("goog equity","px last")


The quotation marks are required.

Single Security with Multiple Fields

You are able to easily copy your formula across multiple fields by locking rows and columns. The single security with a multiple field cell reference, to be entered in cell B2, is:

=BDP (\$A2 , B\$1)


where **\$A2** contains the security and **B\$1** contains the field mnemonic, as shown in Figure 2.



	A	B	C
1		px_last	name
2	goog equity	626.81	

Figure 2: Single Security with Multiple Fields


The **\$** in front of a column or row number locks the column or row. This enables you to drag and copy your formula across multiple cells without having to change it. Figure 3 displays the dragging of a formula.



	A	B	C
1		px_last	name
2	goog equity	626.81	

Figure 3: Dragging the Formula

When you finish dragging the formula, your spreadsheet is automatically updated. Cell C2 contains a now locked updated formula, as shown in Figure 4.



	A	B	C	D
1		px_last	name	
2	goog equity	626.81	GOOGLE INC-CL A	

Figure 4: Relative Formula Copied

To lock the column holding the security, press F4 3x when you are in the formula toolbar. This locks the column.


Since automatic array creation is no longer supported, absolute references with multiple fields are no longer valid. For example, **=bdp("ibm equity, "px last, name, crncy")** is *not* valid syntax.

Multiple Securities with Multiple Fields

You are able to easily copy your formula across multiple fields by locking rows and columns. The single security with a multiple field cell reference, to be entered in cell B2, is:

=BDP(\$A2,B\$1)


where \$A2 contains the security and B\$1 contains the field mnemonic, as shown in Figure 5.



	A	B	C
1		px_last	name
2	goog equity	626.81	
3	ibm equity		
4	dell equity		

Figure 5: Multiple Securities with Multiple Fields


The \$ in front of a column or row number locks the column or row. This enables you to drag and copy your formula across multiple cells without having to change it. You can drag and copy a formula either horizontally or vertically.



	A	B	C
1		px_last	name
2	goog equity	626.81	
3	ibm equity		
4	dell equity		

Figure 6: Dragging the Formula Horizontally

When you finish dragging the formula, your spreadsheet is automatically updated. Cell C2 includes a now locked updated formula, as shown in Figure 7.



	A	B	C	D
1		px_last	name	
2	goog equity	626.81	GOOGLE	INC-CL A
3	ibm equity			
4	dell equity			

Figure 7: Relative Formula Copied

Bloomberg

4

After you drag and copy the formulas vertically, all the cells update and include the formula. Figure 8 shows the results of the dragging and copying, including the formula that is in cell C4.

	A	B	C	D	E
1		px_last	name		
2	goog equity	626.81	GOOGLE INC-CL A		
3	ibm equity	118.15	INTL BUSINESS MACHINES CORP		
4	dell equity	28.695	DELL INC		

Figure 8: Relative Formulas Copied

- To lock the security, press F4 3x when you are in the formula toolbar. This locks the relative column for the security.
- To lock the row with the field mnemonics hit F4 2x, when you are in the formula toolbar. This locks the relative row for the fields.

BDS (Bloomberg Data Set)

BDS (Bloomberg Data Set) returns informational bulk data. It pulls in the same information as the Data Set option in the Data wizard. This replaces the BLPB formula.

Syntax

=BDS (security, field)

The single security, with a bulk data reference, to be entered in cell B2, is:

=BDS (A2,B1)

The formula returns the bulk description and automatically appends the number of rows, as shown in Figure 9.

	A	B	C	D	E	F	G	H
		cie des bulk						
	goog equity	Google Inc. provides a web based search engine through its Google.com website. The Company offers a wide range of search options, including web, image, groups, directory, and news searches.						

Figure 9: BDS Example

The absolute reference of the above example, to be entered in cell B2, is:

=BDS("goog equity", "cie des bulk")

BDH (Bloomberg Data History)

BDH (Bloomberg Data History) returns the historical data for a selected security or set of securities. This is the equivalent of the intraday and historical end of day options in the Data Wizard. It replaces the old BLPI, BLPSH and BLPH formulas.

Syntax

```
=BDH (security, fields, start date, end date,  
[optional argument(s)])
```

See *Appendix A: Optional Arguments for BDH* formulas on page 12 for a list of Optional Arguments.

Required Arguments:

- Security
- Field
- Start date (mm/dd/yyyy)
- End date (mm/dd/yyyy)

Historical Time Series

There are two methods for entering arguments into the Excel spreadsheet, via a range, or via an absolute value.

Specifying optional arguments via a range

Syntax:

```
=BDH(security, field, start date, end date, range  
of optional argument codes, range of values for  
optional arguments)
```

Figure 10 shows how optional arguments are used. In the Figure:

- Cells F1-F3 is the range of optional arguments.
- Cells G1-G3 is the range of VALUES for the optional arguments.




Figure 10: BDH with Optional Arguments

Specifying optional arguments via Absolute Reference

The absolute reference formula for Figure 10 is:

```
=BDH(security, field, Start Date, End date,
"curr=GBP, Days=C, Sort=D")
```

The syntax for the example is:

```
=BDH(security, field, Start Date, End date,
"optional argument = argument value")
```

- You can specify optional arguments in **any** order and advanced options are available. For example, the **Fill** argument has a custom option, where any desired value can be chosen to display for non-trading days. For additional information, see *Appendix A: Optional Arguments for BDH* on page 12.
- Although End Date is a required parameter, if you cell reference a blank cell for this parameter, it defaults to the current date. If you are using an absolute reference, use a double set of quotes for this parameter to achieve the same outcome: "".
- A formula requesting one historical data point is achieved by using the SAME start and end date in the BDH formula.

Intraday Historical Time Series

For historical intraday requests, where you are requesting granular data, use **Barsz** or **BarTp** for the optional parameters. Just as with the previous BLPI formulas, **only a specific set of fields are valid**. These fields include **OPEN, HIGH, LOW, LAST_PRICE, NUMBER_TICKS, VOLUME**




Figure 11: Intraday Historical Time Series with Optional Arguments

In Figure 11, the formula requests an opening tick ask data for 5 minute intervals from 2/22/2007 09:00 to 2/23/2007 16:00.

Syntax

```
=BDH(security, field, start date/time, end
date/time, range of optional arguments, range of
values for optional arguments)
```

Specifying Optional Arguments via a Range:

```
=BDH(security, field, start date& time, end date &
time, F1:F2, G1:G2)
```

Specifying Optional Arguments via a Absolute Referencing

```
=BDH(security, field, start date& time, end date&
time, "barsz=5,bartp=a")
```

Other optional arguments, such as Fill, Orientation and Per, are valid for the intraday requests. For additional information, see *Appendix A: Optional Arguments for BDH* on page 12.

Overrides

Syntax


```
=BDP(Security, fields, input field, input value cell)
```

Arrays will NOT be created automatically, so cell anchoring is necessary.

Cell referencing

```
=BDP($A3,B$2,$C$2,$C$1)
```

Figure 12 details the above formula, where the formula is entered into cell B3.



	A	B	C	D	E
1			100		
2		DUR ASK	PX ASK		
3	bud9 corp	2.387	100		
4					

Figure 12: Cell Referencing Example

The following are true of the formula:

- Locks the column holding the security
- Locks the row holding the fields
- Creates an absolute lock for the input field
- Creates an absolute lock for the input value cell.


Returning a value to one cell

Syntax

**=BDP(security, reactive field, input field(s),
input value cell(s))**

=BDP(A3,B2,C2,C1)

Figure 13 details the above formula, where the formula is entered into cell B3.



The screenshot shows a Bloomberg worksheet with the following data:

	A	B	C	D
1				100
2		DUR ASK	PX ASK	
3	bud9 corp	2.387		
4				
5				

The formula bar for cell B3 shows: **=BDP(A3,B2,C2,C1)**


Figure 13: Returning a Value to One Cell Example

As you only want to produce **one** value, do not drag and copy the formula. It is also not necessary to use cell locking.

Cell referencing with multiple input cells/values

=BDP(\$A3,B2,\$C\$2:\$D\$2,\$C\$1:\$D\$1)

Figure 14 details the above formula, where the formula is entered into cell B3.



The screenshot shows a Bloomberg worksheet with the following data:

	A	B	C	D	E
1			20070220	100	
2		DUR ASK	SETTLE DT	PX ASK	
3	bud9 corp	2.476	02/20/2007	100	
4					
5					
6					

The formula bar for cell B3 shows: **=BDP(\$A3,B2,\$C\$2:\$D\$2,\$C\$1:\$D\$1)**

Figure 14: Cell referencing with Multiple Input cells/values Example

To see the placement of all the cells in the formula, click in cell A3 and press the F2 key. Figure 15 displays the results.

	A	B	C	D	E
1			20070220		100
2		DUR ASK	SETTLE DT	PX ASK	
3	bud9 corp	=BDP(\$A3,B2,\$C\$2:\$D\$2,\$C\$1:\$D\$1)			
4					
5					

Figure 15: Placement of Cells in the Formula

Absolute Referencing

The following section provides examples for absolute referencing.

Index Weight Override Example (the index ticker is case sensitive):

```
=BDP("BUD9 Corp","DUR_ASK","PX_ASK","100")
```

Alternatively you can use the syntax:

```
=BDP(Security, field, input field=input value)
```

One input field/value example:

```
=BDP("BUD9 Corp","DUR_ASK","PX_ASK=100")
```

Using multiple reactive fields is not possible with the literal reference. However, you **can** use multiple input fields and input values.


Two input fields/values Example:

```
=BDP("BUD9 Corp","DUR_ASK","PX_ASK=100,SETTLE_DT=20070102")
```

Which is the same as the below formula with the alternative syntax:

```
=BDP("BUD9 Corp","DUR_ASK","PX_ASK, SETTLE_DT","100,20070220")
```

You can also do a **combination** of absolute and cell referencing to derive the same data as in the above examples, as shown in Figure 16.



The screenshot shows a Bloomberg Excel spreadsheet. The formula bar at the top displays the formula `=BDP(A3,B2,"PX ASK","100")`. The spreadsheet has columns A through E and rows 1 through 5. Cell A3 contains the text "bud9 corp". Cell B2 contains the text "DUR ASK". Cell B3, which is highlighted with a blue border, contains the numerical value "2.387".

	A	B	C	D	E
1					
2		DUR ASK			
3	bud9 corp	2.387			
4					
5					

Figure 16: Combining Absolute and Cell Referencing Examples

Appendix A: Optional Arguments for BDH formulas

Argument	Description/ Possible Argument Values
Curr	<p>Currency</p> <p>Currency of the ISO Code. The 3 letter ISO code, in quotation marks, of the currency (i.e. "USD" for US Dollars).</p> <p>This is NOT available for Intraday.</p>
Days	<p>Non-Trading Days</p> <p>Sets whether to either exclude or include Non-trading days where no data came in.</p> <p>The valid enters are:</p> <ul style="list-style-type: none"> ■ N - All weekdays ■ C - All calendar days ■ T - Trading ■ W - Weekdays ■ A - all days. <p>See also: Fill</p>
Dts	<p>Show Date is the Boolean flag that enables you to turn on/off the trading date.</p> <p>The valid entries are:</p> <ul style="list-style-type: none"> ■ TRUE = Show the date ■ FALSE = Hide the date.
DtFmt	<p>Date output format</p> <p>Allows the you to return a date format as either:</p> <ul style="list-style-type: none"> ■ regular (i.e: yyyyymmdd) ■ relative date format (ie: FQ12006). <p>For Historical end of day only</p>
Fill	<p>Filler value</p> <p>If Days is set to display Non- trading days, then this is the data to return for that day. The default is C.</p> <p>The valid entries are:</p> <ul style="list-style-type: none"> ■ C - Carries over the last available data. ■ P - Carries over the previous day's data. ■ E - Returns an error message. ■ B - Returns a blank. <p>Custom option. Example: Fill= 0 will return 0 for each non trading day</p>

Orientation	<p>Direction/Orientation for how you want the fields to appear in your spreadsheet.</p> <p>Valid entries are:</p> <ul style="list-style-type: none"> ■ H - Horizontal ■ V - Vertical.
Per	<p>Periodicity</p> <p>The periodicity sets the interval within a fixed period of time when data is returned:</p> <p>The base periods are: D (Daily), W (weekly), M (monthly), Q (quarterly), S (semi), Y (Yearly).</p> <p>The Calendar, Fiscal, or Actual modifiers provide greater flexibility when used with the base period. The modifier should precede the base period. Valid modifiers are CD (Calendar Day), CW (Calendar Week), CM (Calendar Month), CQ (Calendar Quarter), CS (Calendar Semi), CY (Calendar Year), FQ (Fiscal Quarter), FS (Fiscal Semi), FY (Fiscal Year), AD (Annual Day), AW (Annual Week), AM (Annual Month), AQ (Annual Quarter), AS (Annual Semi), AY (Annual Year). F should only be used with Q, S, and Y.</p> <p>This is NOT available for Intraday</p>
Points	<p>The number of periods to download from the end date.</p> <p>This is set in periodicity (per) according to Days, Weeks, Months, Quarters, or Years.</p>
QtTyp	<p>Price/Yield Quote</p> <p>This defines historical Price or yield for a yield quoted debt instrument.</p> <p>The valid entries are:</p> <ul style="list-style-type: none"> ■ P - Price ■ Y - Yield.
Quote	<p>Quote Calculation</p> <p>This defines Historical average price or closing prices.</p> <p>The valid entries are:</p> <ul style="list-style-type: none"> ■ G - Golden Mean ■ A - Average ■ C - Close.
BarSz	<p>Bar Size</p> <p>Indicates the interval for data retrieval. Any number between 1 and 1440. Default is 1 minute.</p> <p>For Intraday historical ONLY</p>

BarTp	Bar Type B, Bid, A, Ask, L, Last Only for Intraday
Sort	Order to sort your time or dates. The valid entries are: <ul style="list-style-type: none"> ■ C- Chronological ■ A - Ascend ■ R – Reverse ■ D - Descend.



Help x2

Press the <HELP>
key twice for instant
live assistance.

bloomberg.com

Frankfurt
+49 69 92041 0
Hong Kong
+852 2977 6000

London
+44 20 7330 7500
New York
+1 212 318 2000

San Francisco
+1 415 912 2960
São Paulo
+55 11 3048 4500

Singapore
+65 6212 1000
Sydney
+61 2 9777 8600

Tokyo
+81 3 3201 8900

BLOOMBERG, BLOOMBERG PROFESSIONAL, BLOOMBERG MARKETS, BLOOMBERG NEWS, BLOOMBERG ANYWHERE, BLOOMBERG TRADEBOOK, BLOOMBERG BONDTRADER, BLOOMBERG TELEVISION, BLOOMBERG RADIO, BLOOMBERG PRESS and BLOOMBERG.COM are trademarks and service marks of Bloomberg Finance L.P., a Delaware limited partnership, or its subsidiaries. The BLOOMBERG PROFESSIONAL service (the "BPS") is owned and distributed locally by Bloomberg Finance L.P. (BFLP) and its subsidiaries in all jurisdictions other than Argentina, Bermuda, China, India, Japan and Korea (the "BLP Countries"). BFLP is a wholly-owned subsidiary of Bloomberg L.P. ("BLP"). BLP provides BFLP with all global marketing and operational support and service for these products and distributes the BPS either directly or through a non-BFLP subsidiary in the BLP Countries.